Scientific Computing II, Summer 2012
A Two-Body Model with Gravity Forces

```
> restart;
> with(DEtools):
> with(LinearAlgebra):
> with(plots):
```

Two Point-Masses Connected by a Spring (with mass = 0)

m1 and m2 define the masses of the two bodies

```
> m1 := 1; m2 := 2;
```

$$m1 := 1$$

$$m2 := 2$$

spring force acting along the vector q-r:

```
> spring_ode := {  m1*diff( q[1](t),t,t ) = r[1](t)-q[1](t),
                   m1*diff( q[2](t),t,t ) = r[2](t)-q[2](t),
                   m2*diff( r[1](t),t,t ) = q[1](t)-r[1](t),
                   m2*diff( r[2](t),t,t ) = q[2](t)-r[2](t)};
```

$$spring\_ode := \left\{ 2\left(\frac{d^2}{dt^2} r_1(t)\right) = q_1(t) - r_1(t),\ 2\left(\frac{d^2}{dt^2} r_2(t)\right) = q_2(t) - r_2(t),\ \frac{d^2}{dt^2} q_1(t) = r_1(t) - q_1(t),\ \frac{d^2}{dt^2} q_2(t) = r_2(t) - q_2(t) \right\}$$

We try to let Maple solve this ODE with the initial mass positions as initial conditions - however, this does not lead to a determined solution:

```
> dsolve( { op(spring_ode), q[1](0)=1, q[2](0)=0, r[1](0)=0,
  r[2](0)=1},
                              {q[1](t), q[2](t), r[1](t),
  r[2](t)});
```

$$\left\{ q_1(t) = \frac{1}{3} + \_C6\, t + \_C7 \sin\left(\frac{\sqrt{6}\, t}{2}\right) + \frac{2}{3}\cos\left(\frac{\sqrt{6}\, t}{2}\right),\right.$$
$$q_2(t) = \frac{2}{3} + \_C2\, t + \_C3 \sin\left(\frac{\sqrt{6}\, t}{2}\right) - \frac{2}{3}\cos\left(\frac{\sqrt{6}\, t}{2}\right),$$
$$r_1(t) = -\frac{1}{2}\_C7 \sin\left(\frac{\sqrt{6}\, t}{2}\right) - \frac{1}{3}\cos\left(\frac{\sqrt{6}\, t}{2}\right) + \frac{1}{3} + \_C6\, t,$$
$$\left. r_2(t) = -\frac{1}{2}\_C3 \sin\left(\frac{\sqrt{6}\, t}{2}\right) + \frac{1}{3}\cos\left(\frac{\sqrt{6}\, t}{2}\right) + \frac{2}{3} + \_C2\, t \right\}$$

Two further initial conditions are missing - the velocities (first derivatives):

```
> qrsol := dsolve( { op(spring_ode),
            q[1](0)=1, q[2](0)=0, D(q[1])(0) = 0, D(q[2])(0) =
  1,
            r[1](0)=0, r[2](0)=2, D(r[1])(0) = 0.5, D(r[2])(0)
  = 0 },
```

```
            {q[1](t), q[2](t), r[1](t), r[2](t)}):
> qs1 := subs(qrsol, q[1](t)); qs2 := subs(qrsol, q[2](t));
  rs1 := subs(qrsol, r[1](t)); rs2 := subs(qrsol, r[2](t));
```

$$qs1 := \frac{1}{3} + \frac{t}{3} - \frac{1}{9}\sin\left(\frac{\sqrt{6}\,t}{2}\right)\sqrt{6} + \frac{2}{3}\cos\left(\frac{\sqrt{6}\,t}{2}\right)$$

$$qs2 := \frac{4}{3} + \frac{t}{3} + \frac{2}{9}\sin\left(\frac{\sqrt{6}\,t}{2}\right)\sqrt{6} - \frac{4}{3}\cos\left(\frac{\sqrt{6}\,t}{2}\right)$$

$$rs1 := \frac{1}{18}\sin\left(\frac{\sqrt{6}\,t}{2}\right)\sqrt{6} - \frac{1}{3}\cos\left(\frac{\sqrt{6}\,t}{2}\right) + \frac{1}{3} + \frac{t}{3}$$

$$rs2 := -\frac{1}{9}\sin\left(\frac{\sqrt{6}\,t}{2}\right)\sqrt{6} + \frac{2}{3}\cos\left(\frac{\sqrt{6}\,t}{2}\right) + \frac{4}{3} + \frac{t}{3}$$

```
> qplot := plot( [ qs1,qs2, t=0..10], scaling=CONSTRAINED,
  colour=blue ):
  rplot := plot( [ rs1,rs2, t=0..10], scaling=CONSTRAINED,
  color=red ):
  display(qplot,rplot);
```



Numercal Solution (computed by Maple)

Starting conditions:

(note: to avoid that the total system moves away, the total momentum needs to be = 0)

```
> qrincon :=
  [[q[1](0)=1, q[2](0)=0, D(q[1])(0) = 0, D(q[2])(0) = 1,
    r[1](0)=0, r[2](0)=2, D(r[1])(0) = 0, D(r[2])(0) =
  -0.5]];
```

$$qrincon := [[q_1(0) = 1, q_2(0) = 0, D(q_1)(0) = 0, D(q_2)(0) = 1, r_1(0) = 0, r_2(0) = 2, D(r_1)(0) = 0, D(r_2)(0) = -0.5]]$$

```
> Tend := 4:
  qnum := DEplot(spring_ode, [q[1],q[2],r[1],r[2]],
  t=0..Tend, qrincon,
                 stepsize=0.2, scene=[q[1],q[2]],
  linecolor=blue):
  rnum := DEplot(spring_ode, [q[1],q[2],r[1],r[2]],
```

```
t=0..Tend, qrincon,
                    stepsize=0.2, scene=[r[1],r[2]],
linecolor=red):
display(qnum,rnum);
```



Two-Body Problem with Gravity

```
> m1 := 1; m2 := 6;
```

$$m1 := 1$$

$$m2 := 6$$

We define a function to compute the gravity force:

```
> Fg1 := (q1,q2,r1,r2) -> -m1*m2/( (q1-r1)^2 + (q2-r2)^2
  )^(3/2)
                    *(q1-r1);
  Fg2 := (q1,q2,r1,r2) -> -m1*m2/( (q1-r1)^2 + (q2-r2)^2
  )^(3/2)
                    *(q2-r2);
```

$$Fg1 := (q1, q2, r1, r2) \rightarrow -\frac{m1\ m2\ (q1 - r1)}{((q1 - r1)^2 + (q2 - r2)^2)^{(3/2)}}$$

$$Fg2 := (q1, q2, r1, r2) \rightarrow -\frac{m1\ m2\ (q2 - r2)}{((q1 - r1)^2 + (q2 - r2)^2)^{(3/2)}}$$

q and r are the position vectors of the two bodies (in 2D);
the gravity force (2nd derivative of q and r) is always in the direction of the vector q-r

```
> grav_ode := {  m1*diff( q[1](t),t,t ) =
  Fg1(q[1](t),q[2](t),r[1](t),r[2](t)),
                     m1*diff( q[2](t),t,t ) =
  Fg2(q[1](t),q[2](t),r[1](t),r[2](t)),
                     m2*diff( r[1](t),t,t ) =
  Fg1(r[1](t),r[2](t),q[1](t),q[2](t)),
                     m2*diff( r[2](t),t,t ) =
  Fg2(r[1](t),r[2](t),q[1](t),q[2](t)) };
```

$$grav\_ode := \left\{ 6\left(\frac{d^2}{dt^2} r_1(t)\right) = -\frac{6\,(r_1(t) - q_1(t))}{((r_1(t) - q_1(t))^2 + (r_2(t) - q_2(t))^2)^{(3/2)}}, \right.$$

$$6\left(\frac{d^2}{dt^2} r_2(t)\right) = -\frac{6\,(r_2(t) - q_2(t))}{((r_1(t) - q_1(t))^2 + (r_2(t) - q_2(t))^2)^{(3/2)}},$$

$$\frac{d^2}{dt^2} q_1(t) = -\frac{6\,(q_1(t) - r_1(t))}{((q_1(t) - r_1(t))^2 + (q_2(t) - r_2(t))^2)^{(3/2)}},$$

$$\left. \frac{d^2}{dt^2} q_2(t) = -\frac{6\,(q_2(t) - r_2(t))}{((q_1(t) - r_1(t))^2 + (q_2(t) - r_2(t))^2)^{(3/2)}} \right\}$$

Initial conditions:

```
> qrincon :=
  [[q[1](0)=5, q[2](0)=0, D(q[1])(0) = 0.2, D(q[2])(0) = 1,
    r[1](0)=0, r[2](0)=1, D(r[1])(0) = 0.2, D(r[2])(0) =
  -1/6]];
```

$$qrincon := [[q_1(0) = 5, q_2(0) = 0, D(q_1)(0) = 0.2, D(q_2)(0) = 1, r_1(0) = 0, r_2(0) = 1,$$

$$D(r_1)(0) = 0.2, D(r_2)(0) = \frac{-1}{6}]]$$

And we will need a numerical methods to compute this rather complicated ODE:

```
> qnum := DEplot(grav_ode, [q[1],q[2],r[1],r[2]], t=0..100,
  qrincon,
                     stepsize=0.2, scene=[q[1],q[2]],
  linecolor=blue):
  rnum := DEplot(grav_ode, [q[1],q[2],r[1],r[2]], t=0..100,
  qrincon,
                     stepsize=0.2, scene=[r[1],r[2]],
  linecolor=red):
  display(qnum,rnum);
```



Solution in a Local Coordinate System

We try to represent the solution as it is seen from the point-of-view of the two bodies (example: movement of the Moon as seen from the Earth).
F mor that purpose, we let qsolve generate the numerical solution as a procedure that we can call later:

```
> qrsol := dsolve( { op(grav_ode), op(qrincon[1]) }, numeric,
          {q[1](t), q[2](t), r[1](t), r[2](t)}, output=listprocedure);
```

$$qrsol := \Big[t = (\textbf{proc}(t) \; ... \; \textbf{end proc}), q_1(t) = (\textbf{proc}(t) \; ... \; \textbf{end proc}),$$
$$\frac{d}{dt} q_1(t) = (\textbf{proc}(t) \; ... \; \textbf{end proc}), q_2(t) = (\textbf{proc}(t) \; ... \; \textbf{end proc}),$$
$$\frac{d}{dt} q_2(t) = (\textbf{proc}(t) \; ... \; \textbf{end proc}), r_1(t) = (\textbf{proc}(t) \; ... \; \textbf{end proc}),$$

$$\frac{d}{dt}r_1(t) = (\mathbf{proc}(t)\ \ \ldots\ \ \mathbf{end\ proc}),\ r_2(t) = (\mathbf{proc}(t)\ \ \ldots\ \ \mathbf{end\ proc}),$$

$$\left.\frac{d}{dt}r_2(t) = (\mathbf{proc}(t)\ \ \ldots\ \ \mathbf{end\ proc})\right]$$

Maple generates a procedure for each involved position function (plus for its derivatives) -> we'll only need the position functions:

```
> qs1 := subs(qrsol, q[1](t)): qs2 := subs(qrsol, q[2](t)):
  rs1 := subs(qrsol, r[1](t)): rs2 := subs(qrsol, r[2](t)):
> qplot := plot( [ qs1,qs2, 0..100], scaling=CONSTRAINED,
  colour=blue, thickness=2 ):
  rplot := plot( [ rs1,rs2, 0..100], scaling=CONSTRAINED,
  color=red, thickness=2 ):
  display(qplot,rplot);
```



The relative motion of the two bodies is now obtained as the difference function of the positions
(note that the result is not just a circular motion around the point of origin):

```
> ds1 := qs1-rs1; ds2 := qs2-rs2;
  plot( [ ds1,ds2, 0..100], scaling=CONSTRAINED);
```

$$ds1 := qs1 - rs1$$

$$ds2 := qs2 - rs2$$



4
2
-4
-2
0
2
4
-2
-4
-6

>